
```

LIST (FIRST=n, LAST=n, PREFIX=名前) リスト名 [=] 変数リスト ;
        あるいは
LIST(DROP) リスト名 変数名のリスト ; あるいは
LIST(PRINT, DELETE) リスト名のリスト ;

```

機能 :

LIST は TSP 変数リストに 1 つの名前を与え、後でプログラムで変数リストが必要になった時にいつでも使用できるようにします。このコマンドは、変数の長いリストを繰り返し使用するとき便利です。リストを定義した後、(方程式コード内を除いて) コマンド内で何回も使え、リストの内容をリスト名で置き換えます。リストに添え字を付けたり、ラグを付けることもできます。PROC の引数もリストとして扱われます。

使用法:

LIST 名は規則上正しい TSP の変数名であればよく、この LIST に続いて変数名 (ラグ付変数やリード変数を含む) が与えられます。また変数でなくとも他の LIST 名も可能です。

LIST は入れ子 (nest) も可能です。すなわち、LIST に続く変数名に他の LIST が含まれてもよいのです。その長さに関する唯一の制限は、変数の最終的なリストの長さです。LIST は再帰的に定義する事は出来ません。正しくない再帰的な例は、(もし LL が既にリストとして定義されていない場合は) LIST LL A B LL ; です。

変数の LIST は、VAR1-VAR20 のように範囲として設定する事ができます。このような LIST は VAR1, VAR2, ..., VAR19, VAR20 として解釈されます。このような範囲指定は、定数 (1-20) やラグ付変数 (X(-1)-X(-20)) にも使えます。次の DOT コマンドで分かるように、LIST はコマンドの中でも直接使えます。LIST を DOT ループの中で使って、LIST の各要素を便利に操作することもできます。

```

LIST VARLIST VAR1-VAR20 ;
DOT 1-20 ;
    ----- VAR. に関する計算 -----
ENDDOT ;
    ----- VARLIST の操作 -----

```

この様にして (リスト名は GENR, SET あるいは方程式の指定では許されていないので) 変数をグループとしたり、GENR で個々の変数として扱うことができます。

オプション:

DROP/NODROP は現在のリストから変数を落とします。

FIRST= リスト内の名前の系列の開始番号。既定値は 1。

LAST= リスト内の名前の系列の最終番号。既定値は 1。

PREFIX= リストを作成する接頭子に用いる名前; namefirst–namelast. 例を参照。

例:

いくつかの例

LIST

```
LIST EQ EQ1-EQ4 ;
LIST INSTVAR C TIME LM G ;
LIST ENDOGVAR GNP CONS I R LP ;
LIST ALLVARS INSTVAR ENDOGVAR ;
LIST LAGS X(-1)-X(-20) ;
LIST STATES 1-50 ;
```

リスト ALLVARS は系列 C, TIME, LM, G, GNP, CONS, I, R, LP からなります。

次の例は、繰り返しタイプするのを避けるために DOT ループと組み合わせたときに、如何に強力に使えるかを示します。

```
PRINT PAT72-PAT76 RND72-RND76 ;
PARAM A72-A76 DELT72-DELT76 BETA;
DOT 72-76;
  FRML EQ. PAT. = EXP(A. + BETA*RND.+ DELT72*RND72 + DELT73*RND73
    + DELT74*RND74 + DELT75*RND75 + DELT76*RND76) ;
  PARAM A. 1.0 DELT. ;
ENDDOT;
LSQ (NOPRINT, STEP=BARDB) EQ72-EQ76 ;
```

この例では、パネルデータ・モデル (5 年間の数百単位の個々のデータ) に関する系列、方程式、そしてパラメータは、各年次の変数を繰り返しタイプするのを避けるために LIST 名によってすべて参照することができます。明らかに、LIST は数学的な表現にならないので方程式の中では使えません。しかし応用によっては、良く似た方程式を再生するのに DOT ループのなかで EQSUB を使うと便利です。

実行するまで LIST 内の変数の数が分からないとします。この場合、オプションを使って変数の長さを求めることができます。

```
BEGYR=72; ENDYR=76;
LIST(PREFIX=EQ,FIRST=BEGYR,LAST=ENDYR) EQS;
```

これによって、EQ72, EQ73, EQ74, EQ75, EQ76 という EQS のリストができます。

次は、もっと単純な例です。

LIST(FIRST=5, LAST=1) DECR;	? 5 4 3 2 1
LIST(DROP) DECR 3 2;	? 5 4 1
LIST(PREFIX=B) BS X Y Z(-1);	? BX BY BZ(-1)
LIST BSMID BS(2);	? Y (添え字リスト)
PRINT BS(-1);	? X(-1) Y(-1) Z(-2) (ラグ付リスト)
LIST(PRINT) BS;	? X Y Z(-1) の数値でなく名前をプリントする
LIST(DELETE) BS;	? BS のリスト定義を削除する

事実、リストは任意の一連のワードあるいは引数を保存するのに用いることができます。これは変数名を含まないが、TSP が理解できるリストのいくつかの例です。

LIST STRING "A title string";	
TITLE STRING;	? TITLE "A title string"; と同じ
LIST OPTS MEAN=5;	
RANDOM(OPTS) X;	? RANDOM(MEAN=5) X; と同じ

アウトプット:

LIST にはアウトプットはありません。TSP 変数 LIST はデータ領域に保存されます。SHOW LIST ; によって現在定義されているリストを見ることができます。